# SPDX-License-Identifier: BSD-3-Clause
# Copyright (c) 2026, The OpenROAD Authors

load("@openroad-pip//:requirements.bzl", "requirement")
load("@rules_python//python:defs.bzl", "py_test")
load("//bazel:man_pages.bzl", "man_pages")

man_pages(
    name = "man_pages",
    docs_srcs = [
        "Makefile",
        "src/man_fontfix.tmac",
        # md/man1 holds the only checked-in markdown. md/man2 and md/man3 are
        # written by this rule's own preprocess step, so globbing them would
        # declare the action's outputs as its inputs: the input set would vary
        # with leftover files from previous builds, and rewriting them trips
        # --guard_against_concurrent_changes ("modified during execution").
    ] + glob(["md/man1/**/*.md"]),
    messages = [
        # The ORD messages, from src/ itself.
        "//:messages_txt",
        "//src/ant:messages_txt",
        "//src/cgt:messages_txt",
        "//src/cts:messages_txt",
        "//src/cut:messages_txt",
        "//src/dbSta:messages_txt",
        "//src/dft:messages_txt",
        "//src/dpl:messages_txt",
        "//src/drt:messages_txt",
        "//src/dst:messages_txt",
        "//src/est:messages_txt",
        "//src/exa:messages_txt",
        "//src/fin:messages_txt",
        "//src/gpl:messages_txt",
        "//src/grt:messages_txt",
        "//src/gui:messages_txt",
        "//src/ifp:messages_txt",
        "//src/mpl:messages_txt",
        "//src/odb:messages_txt",
        "//src/pad:messages_txt",
        "//src/par:messages_txt",
        "//src/pdn:messages_txt",
        "//src/ppl:messages_txt",
        "//src/psm:messages_txt",
        "//src/ram:messages_txt",
        "//src/rcx:messages_txt",
        "//src/rmp:messages_txt",
        "//src/rsz:messages_txt",
        "//src/stt:messages_txt",
        "//src/syn:messages_txt",
        "//src/tap:messages_txt",
        "//src/tst:messages_txt",
        "//src/upf:messages_txt",
        "//src/utl:messages_txt",
        "//src/web:messages_txt",
    ],
    readmes = [
        "//src/ant:README.md",
        "//src/cgt:README.md",
        "//src/cts:README.md",
        "//src/dft:README.md",
        "//src/dpl:README.md",
        "//src/drt:README.md",
        "//src/est:README.md",
        "//src/exa:README.md",
        "//src/fin:README.md",
        "//src/gpl:README.md",
        "//src/grt:README.md",
        "//src/gui:README.md",
        "//src/ifp:README.md",
        "//src/mpl:README.md",
        "//src/odb:README.md",
        "//src/pad:README.md",
        "//src/par:README.md",
        "//src/pdn:README.md",
        "//src/ppl:README.md",
        "//src/psm:README.md",
        "//src/ram:README.md",
        "//src/rcx:README.md",
        "//src/rmp:README.md",
        "//src/rsz:README.md",
        "//src/stt:README.md",
        "//src/syn:README.md",
        "//src/tap:README.md",
        "//src/upf:README.md",
        "//src/utl:README.md",
        "//src/web:README.md",
    ],
    scripts = [
        "//docs/src/scripts:link_readmes.sh",
        "//docs/src/scripts:md_roff_compat.py",
        "//docs/src/scripts:manpage.py",
        "//docs/src/scripts:extract_utils.py",
    ],
    visibility = ["//visibility:public"],
)

# Replicates the ReadTheDocs Sphinx build locally (hermetic Python deps).
# See: https://github.com/The-OpenROAD-Project/OpenROAD/issues/9885
#
# Usage:
#   bazelisk test //docs:sphinx_build_test
#   bazelisk test --test_tag_filters=doc_check //docs/...
py_test(
    name = "sphinx_build_test",
    srcs = ["sphinx_build_test.py"],
    data = glob(
        ["**/*"],
        exclude = [
            "BUILD.bazel",
            "sphinx_build_test.py",
            "_build/**",
            "_readthedocs/**",
            "main/**",
            # Everything docs/.gitignore lists as intermediate output: the man
            # page build (//docs:man_pages) writes these into the source tree,
            # so globbing them would make this test's inputs -- and so its
            # cache key and result -- depend on whether man pages happen to
            # have been built here. conf.py excludes all four from the Sphinx
            # build anyway.
            "cat/**",
            "html/**",
            "man/**",
            "md/man2/**",
            "md/man3/**",
            "build/**",
            "src/test/results/**",
        ],
    ) + [
        # Staged at the runfiles workspace root so conf.py's setup(app)
        # can resolve ../README.md and main/README.md through the
        # `main -> ..` symlink it creates.
        "//:README.md",
    ],
    tags = [
        "doc_check",
    ],
    deps = [
        requirement("sphinx"),
        requirement("sphinx-book-theme"),
        requirement("sphinx-copybutton"),
        requirement("sphinx-external-toc"),
        requirement("sphinx-tabs"),
        requirement("sphinxcontrib-mermaid"),
        requirement("myst-parser"),
        requirement("sphinx_llm"),
        requirement("standard-imghdr"),
    ],
)